
1. Цель и задачи курсового проекта
Выполняя курсовую работу по функциональному и логическому программированию, студент приобретает навыки самостоятельного решения прикладных задач с использованием декларативной парадигмы программирования. Эти задачи решаются комплексно, во взаимной увязке. Используется материал ранее изученных дисциплин. 
2. Задание на работу
Задание на работу состоит из двух частей:
1. Разработка прототипа экспертной системы по классификации объектов некой предметной области на языке Prolog.
2. Разработка программы по сбору и анализу объектов некой предметной области на языке Haskell.
Предметная область выбирается в соответствии с индивидуальным заданием, получаемым у преподавателя. Ниже перечислен список предлагаемых предметных областей:
1. университет
2. поликлиника
3. кафе
4. отпуск
5. домашнее животное
6. компьютер
7. мобильный телефон
8. школа
9. психологическое тестирование
10. кафедра
11. компьютерные игры
12. автомобиль
13. кинотеатр
14. магазин
15. банк
16. образовательный канал
17. социальная сеть
18. грибы
19. отель
20. экскурсия
21. программное обеспечение
[bookmark: _GoBack]22. ремонт оборудования
23. железнодорожный транспорт
24. онлайн-заказ
25. музей
26. концерт
27. косметика
28. книги
29. воздушный транспорт
30. городской транспорт
3. Реализация экспертных систем на языке Prolog
Экспертная система — компьютерная программа, которая, как и любая другая программа, получает какие-то данные на вход и формирует другие данные на выходе. Особенность в том, что целью работы экспертной системы является замена эксперта (человека) в некоторой предметной области. Естественно, работают такие программы без магии, а по вполне определенному алгоритму. От некоторых других видов интеллектуальных систем они отличаются большим объемом знаний эксперта, необходимым для принятия решения. Например, система, которая контролирует температуру в мангале при жарке шашлыка не может называться экспертной системой.
Экспертные системы создаются для самых различных предметных областей, например:
- созданы системы медицинской диагностики, которые по набору симптомов назначают анализы, по результатам которых ставится диагноз и определяется курс лечения;
- экспертные системы могут следить за соблюдением правил дорожного движения и даже оценивать военную безопасность государства.
Так или иначе, в процессе использования в систему попадают факты пользователя, чаще всего это происходит в форме диалога, при этом система формирует вопросы исходя их данных в базе и данных ранее ответов. Например, система Akinator, угадывающая задуманного пользователем персонажа мультфильма, может спросить «является ли ваш персонаж человеком?».
Важно, что все экспертные системы, вне зависимости от предметной области, имеют схожую структуру — на самом деле, какая разница «угадывает» система диагноз по симптомам или персонажа мультфильма по его параметрам? По этой причине для разработки часто используются специальные инструменты создания типовых систем или заготовки систем, в которые эксперту нужно лишь внести данные и правила, однако, в этой статье мы посмотрим, как сделать простейшую экспертную систему на языке Prolog.
На рисунке ниже схематично приведена структура такой системы. В некоторых системах встречается модуль объяснения решения, который поясняет пользователю результаты работы (почему выставлен именно такой диагноз, например). Иногда, система обучается в процессе диалога с пользователем, при этом в базу данных заносятся новые факты.

Важно понимать, что база данных содержит факты, а база знаний — правила вывода решений на основе фактов и введенных пользователем данных. Правила представляют собой, по сути, исходный код программы, — именно поэтому для разработки экспертных систем удобно использовать языки типа SWI Prolog, ведь они позволяют добавлять новые правила в программу прямо во время ее исполнения.
Ниже приведен пример разработки сильно упрощенной экспертной системы. Рассмотрим систему, помогающую студенту выбрать тему дипломной работы.
Так как в основе любой экспертной системы находится база данных, то разработка начинается с анализа предметной области. Тут надо понять какие критерии стоит учитывать при подборе темы дипломной работы. Темы выпускных работ обычно формируют преподаватели исходя из своих интересов. Студент выбирает тему бакалаврской работы из набора имеющихся с учетом:
· желания работать с преподавателем, выдающим тему.
· своих интересов;
· сложности темы (по экспертной оценке руководителя темы);
· своих знаний по теме работы;
· владению необходимыми технологиями если они указаны руководителем как обязательные.
Например, если студент при обучении у преподавателя Пупкина с трудом получил тройку — маловероятно он пойдет к нему на дипломирование. Если студент ничего не знает в области блокчейна, но ему хотелось бы развиваться в этой сфере — возможно он выберет соответствующую тему. Тема работы «Разработка клиент-серверного приложения для игры в покер» может быть интересна студенту, но он не возьмет ее если руководитель требует писать сервер на Erlang, а студент не любит функциональное программирование.
Таким образом, предметная область задачи состоит из сущностей:
· руководитель ВКР;
· область интересов;
· тема ВКР;
· технология выполнения;
· студент.
Модель предметной области с использованием нотации диаграммы классов UML приведена на рисунке:

Исходя из информации, изложенной в обзоре предметной области, ясно, что хранить программа должна темы и данные студента. Однако, данные студента вводятся в процессе диалога с ним и не должны храниться после чтобы не влиять на работу с системой следующего пользователя. Таким образом, в базе надо хранить лишь информацию о темах.
Для наполнения базы были найдены темы ВКР, по первому запросу поисковая система выдала информацию о темах Высшей школы экономики. Однако, опубликованный список тем содержит лишь информацию о названии и руководителе, остальная информация была сформирована мной и не является точной — данные в базу данных экспертной системы должен вносить эксперт (руководитель ВКР).
В результате создана такая база данных:
theme('Повышение эффективности операций в цепях поставок', 
'Бажина Д.Б.', 
complex(70), 
knowledge_areas(['экономика', 'управление', 'экономика', 'оптимизация']),
skills(['1C'])
).
theme('Повышение эффективности управления цепями поставок скоропортящихся товаров', 
'Бажина Д.Б.', 
complex(75), 
knowledge_areas(['экономика', 'управление', 'законы', 'экономика', 'оптимизация']),
skills([])
).
theme('Разработка мероприятий по повышению лояльности потребителей транспортных услуг', 
'Бородулина С.А.', 
complex(85), 
knowledge_areas(['транспорт', 'психология']),
skills([])
).
theme('Повышение эффективности управления материальными ресурсами транспортной компании', 
'Бородулина С.А.', 
complex(60), 
knowledge_areas(['транспорт', 'управление', 'экономика', 'оптимизация']),
skills(['1C'])
).
theme('Разработка рекомендаций по применению геоинформационных технологий для повышения качества работы транспортной системы мегаполиса', 
'Бочкарев А.А.', 
complex(60), 
knowledge_areas(['ГИС', 'транспорт', 'оптимизация']),
skills(['1C'])
).
theme('Применение аппарата теории массового обслуживания для повышения эффективности синтеза обслуживающих систем', 
'Булатов М.А.', 
complex(95), 
knowledge_areas(['математика', 'оптимизация']),
skills(['теория массового обслуживания'])
).
На этом этапе можно лишь перебрать все записи базы и вывести их запросом:
?- theme(Theme, Name, Complex, knowledge_areas(Areas), skills(Skills)).
Выбор руководителей
Чтобы посоветовать тему, экспертная система должна получить информацию о студенте. Для начала можно узнать с какими преподавателями он готов работать. Для этого соберем список из всех преподавателей и выведем ему их под номерами. Пусть студент введет номера подходящих ему руководителей.
Собрать всех преподавателей можно так:
findall(Name, theme(_Theme, Name, _Complex, _Areas, _Skills), Names),
list_to_set(Names, Set).
Встроенный предикат findall находит все решения предиката, переданного вторым аргументом и формирует из решений список. Однако, имена в таком списке могут повторяться — если у одного преподавателя было несколько тем, то интерпретатор найдет его имя несколько раз. Поэтому следом вызывается предикат 
list_to_set
который убирает из списка повторы.

Теперь надо вывести эти имена списком так, чтобы к каждому имени был привязан номер. Это позволит пользователю не вводя имя (в котором легко сделать опечатку) выбрать нужные имена. Для этого описан предикат, принимающий на вход список и начальный номер и выводящий элементы списка в столбик так, что у каждого следующего элемента номер на единицу больше. Также написан предикат, выполняющий запрос всех имен (как показано выше) и выводящий их:
print_enumeration_list([], _Number):-!.
print_enumeration_list([Head|Tail], Number):-
write('\t'), write(Number), write(' - '), write(Head), nl,
Next is Number+1,
print_enumeration_list(Tail, Next).
get_names(Names):-
findall(Name, theme(_Theme, Name, _Complex, _Areas, _Skills), Names),
list_to_set(Names, Set).
Результат выполнения запроса:

Теперь можно запросить у пользователя список имен и выбрать соответствующих номерам преподавателей:
nths_1(All, Indexes, Selected):-
Indexes = [], !, Selected = [];
Indexes = [HeadIndex|TailIndexes],
nth1(HeadIndex, All, HeadSelected),
nths_1(All, TailIndexes, TailSelected),
Selected = [HeadSelected|TailSelected].
select_names(SelectedNames):-
get_names(Names),
print_enumeration_list(Names, 1),
write('Введи подходящих преподавателей [номера через запятую]: '),
read(Numbers),
nths_1(Names, Numbers, SelectedNames), !;
% введены неверные данные:
write('Неправильный ввод, повторим...'), nl,
select_names(SelectedNames).
Тут предикат 
select_names
после вывода списка преподавателей запрашивает список номеров и передает их в предикат 
nths_1
который выбирает элементы списка, соответствующие списку номеров (нумерация начинается с единицы). если введены неверные данные — то выведется сообщение об ошибке и процесс повторится (рекурсивно):

Выбор области знаний
Теперь, когда выбраны предполагаемые — можно собрать список областей знаний в которых у них есть темы и выдать их в виде такого же списка.
Сначала можно получить список всех областей, в которых есть темы у выбранных руководителей:
area_of_masters(Names, Area):-
member(Name, Names), 
theme(_Theme, Name, _Complex, knowledge_areas(Areas), _Skills),
member(Area, Areas).
этот предикат сначала перебирает руководителей с помощью встроенного предиката 
member
затем перебирает все темы этого руководителя в базе, для каждой из них получает список областей. Искомая область принадлежит этому списку. Перебрать все области опять помогает 
member
Теперь можно собрать все эти области в список и убрать из него повторы:
collect_areas(Names, Areas):-
findall(Area, 
area_of_masters(Names, Area),
AreasList
),
list_to_set(AreasList, Areas).
Мы смогли получить по списку имен руководителей список областей, в которых они работают без повторов, выведем этот список на экран чтобы студент смог выбрать интересующие его области:
select_areas(Names, SelectedAreas):-
collect_areas(Names, Areas),
print_enumeration_list(Areas, 1),
write('Введи подходящих области [номера через запятую]: '),
read(Numbers),
nths_1(Areas, Numbers, SelectedAreas), !;
write('Неправильный ввод, повторим...'), nl,
select_areas(Names, SelectedAreas).
В настоящий момент можно написать такую цель:
?- select_names(Names), select_areas(Names, Areas).
с ее помощью мы сначала просим выбрать имена, а затем области. Результат ее выполнения:

Указание компетенций студента
Аналогичным образом можно запросить у пользователя набор его компетенций. Для этого можно описать правила поиска всех компетенций, которые требуются в темах выбранных преподавателей по выбранным областям, примерно так:
skills_of_such_themes_that(Names, Areas, Skill):-
theme(_Theme, Name, _Complex, knowledge_areas(ThemeAreas), skills(Skills)),
member(Name, Names),
member(Area, Areas), member(Area, ThemeAreas),
member(Skill, Skills).
collect_skills(Names, Areas, Skills):-
findall(Skill, 
skills_of_such_themes_that(Names, Areas, Skill),
SkillsList
),
list_to_set(SkillsList, Skills).
Дадим пользователю выбрать навыки из имеющегося списка:
select_skills(Names, Areas, SelectedSkills):-
collect_skills(Names, Areas, Skills),
print_enumeration_list(Skills, 1),
write('Введи имеющиеся компетенции [номера через запятую]: '),
read(Numbers),
nths_1(Skills, Numbers, SelectedSkills), !;
write('Неправильный ввод, повторим...'), nl,
select_skills(Names, Areas, SelectedSkills).
Дальше надо убрать из выдачи темы, для выполнения которых у студента не хватает компетенций…
Выбор темы
Для выбора тем, соответствующих всем выбранным критериям написан такой предикат:
matched_theme(Names, Areas, Skills, Theme):-
theme(Theme, Name, _Complex, knowledge_areas(ThemeAreas), skills(ThemeSkills)),
member(Name, Names),
member(Area, Areas), member(Area, ThemeAreas),
\+ (
member(Skill, ThemeSkills),
\+ member(Skill, Skills)
).
Итак, он перебирает все темы в базе данных и для каждой из них проверяет следующее:
· имя преподавателя есть в списке 
Names
· среди областей знаний темы есть тема из выбранного пользователем списка (проверяется с помощью двух вызовов member — выбери такой 
Area
 в списке 
Areas
, что такой же 
Area
 есть в списке 
ThemeAreas
);
· проверяет что НЕТ такого навыка в теме, что им НЕ владеет студент.
Для реализации последней части используется два отрицания. Так:
\+ member(Skill, Skills)
проверяет что навыка Skill нет в списке Skills. Список Skills — навыки студента.
А тут:
\+ (
member(Skill, ThemeSkills),
\+ member(Skill, Skills)
).
первый оператор 
\+
стоит перед группой (почти лямбда-функцией) и завершится успешно если вся группа «провалится». Группа провалится если среди навыков темы найдется такой Skill, что его нет у студента. Таким образом весь приведенный в последнем листинге фрагмент завершится если у студента есть все необходимые навыки для выполнения темы.
Все описанные выше предикаты собраны (вызываются) в предикате выбора темы так:
selected_theme(Theme):-
select_names(Names),
( 
Names = [], write('Ничем не могу помочь');
select_areas(Names, Areas),
( 
Areas = [], write('Ничем не могу помочь');
select_skills(Names, Areas, Skills),
matched_theme(Names, Areas, Skills, Theme)
)
).
То есть если на каком то этапе оказалось, что система не может помочь — например, нет преподавателя, с которым хотел бы работать студент — то процесс подбора темы прерывается с сообщением о проблеме.
Результат подбора темы:

На приведенном рисунке видно, что одна и та же тема выведена дважды. Чтобы убрать это — все названия помещаются в список, который преобразуется во множество:
run(Theme):-
findall(T, selected_theme(T), Themes),
( 
Themes = [], write('Нет подходящих тем');
list_to_set(Themes, ThemesSet),
member(Theme, ThemesSet)
).
Видно, что для создания экспертной системы надо сначала проанализировать предметную область и понять какие данные стоит сохранять в системе, а также определиться с набором правил, по которым система может принимать решения. Затем, стоит описать данные и правила на языке программирования. В созданной тут системе нет никакого интерфейса для эксперта, таким образом система является статической.

4. Разработка программного обеспечения для анализа объектов предметной области
Язык Haskell (Хаскелл) – совокупность символов для обработки и машинного кодирования списков и вычислений. Эта довольно распространенная, чистая и функциональная система получила свое название в честь математика Карри Хаскелла.
Haskell сейчас используется чаще всего по нескольким основным направлениям. Достаточно часто Haskell применяют в финансовом секторе — крупные банки и другие компании, работающие на этом рынке, применяют его для разработки собственных инструментов. Бизнес применяет именно Haskell из-за гарантированности точности вычислений без ошибок.
Haskell часто используют для написания инструментов для обработки текстов, синтаксического анализа и создания фильтр-систем для обработки спама.
Разработчики также применяют Haskell для создания инструментов для тестирования кода на других языках программирования. Здесь используется примерно тот же алгоритм, что и в предыдущем пункте, поскольку Haskell можно легко применять не только для анализа обычных языков, но и языков программирования.
При этом Haskell можно применять и для веб-разработки, он может компилироваться в JavaScript для запуска в браузере, а на сервере запускаться в качестве быстрого машинного кода, который легко обрабатывает десятки тысяч одновременных подключений.
Следующие особенности Haskell делают его достаточно удобным:
· Безопасное управление памятью, сборка мусора
· Компиляция в нативный код
· Статическая типизация, развитая система типов
· Чистые функции
· Ленивые вычисления
· Многопоточное программирование
· Метапрограммирование (генерация кода)
Основу языка составляют функции. Объявление функции состоит из имени функции и списка аргументов вместе с ее выводом. Определение функции – это то, где вы на самом деле определяете функцию.
Например,
add :: Integer -> Integer -> Integer   --объявление функции 
add x y =  x + y                       --определение функции
В Haskell нет встроенных циклических конструкций, как и в Prolog. 
Вычисление факториала можно представить в Haskell:
fact :: Int ->Int 
fact n = fact1 n 0 1
fact1: Int ->Int-> Int ->Int 
fact1 n x f =f.
fact n c x = fact1 n c1 f1
   where 
		c1=c+1
f1=c1*x
Одним из основных типов данных является список. Список — это упорядоченная последовательность элементов, где слово упорядоченная означает, что порядок перечисления элементов имеет значение. 
В Haskell используются две фундаментальные структуры, предназначенные для работы с множественными значениями: списки и кортежи. И те, и другие собирают несколько значений в одно.
В Haskell все списки строятся путём присоединения элементов к пустому списку, []. Запись с квадратными скобками и запятыми - всего лишь синтаксический сахар. Так что [1,2,3,4,5] - абсолютно то же самое, что 1:2:3:4:5:[]
Списки могут содержать что угодно — если всё это что угодно одного и того же типа. Списки подходят под что угодно, так что они могут содержаться в списках. 
Стоит отметить, что тип списка, к примеру, целых чисел отличен от типа целых чисел. Списки различных типов не могут быть присоединены друг к другу, но пустой список можно присоединить к любому списку.
Хранение данных в Haskell можно реализовать с использованием АДТ.
АТД, или Алгебраические Типы Данных (англ. ADT, Algebraic Data Type), занимают почётное место в мире типов Haskell. Абсолютно подавляющее большинство ваших собственных типов будут алгебраическими, и то же можно сказать о типах из множества Haskell-пакетов. Алгебраическим типом данных называют такой тип, который составлен из других типов. Мы берём простые типы и строим из них, как из кирпичей, типы сложные, а из них — ещё более сложные. Это даёт нам невероятный простор для творчества.
data IPAddress = IPAddress String
Тип IPAddress использует один-единственный конструктор значения, но имена типа и конструктора совпадают, конструктор в данном случае унарный потому что теперь он связан с одним значением типа String. И вот как создаются значения типа IPAddress:
  let ip = IPAddress "127.0.0.1"
Значение ip типа IPAddress образовано конструктором и конкретным значением некоего типа:
  let ip = IPAddress      		 "127.0.0.1"
           конструктор    		 значение
           значения       		 типа
           типа IPAddress 		 String
         └     значение типа IPAddress         ┘
Можно задавать n-арные конструкторы, добавляя соответствующее количество значений, содержащихся в типе.
На Haskell вам необходимо написать программу, реализующую один из вариантов анализа объектов, представленных в базе данных. К вариантам таких анализов могут относиться:
- корреляционный анализ,
- регрессионный анализ,
- дисперсионный анализ,
- кластерный анализ,
- анализ вариационного ряда распределения.
Математическую составляющую этих вариантов анализа вы должны были разбирать в курсе математической статистики.
5. Объем работы и ее оформление
Программная часть курсовой работы по функциональному и логическому программированию состоит из реализованных программ, выполненной в соответствии с заданием.
Текстовая часть курсовой работы - “Пояснительная записка” содержит текст, оформленный в соответствии с ГОСТ 7.32-2017 ОТЧЕТ О НАУЧНО-ИССЛЕДОВАТЕЛЬСКОЙ РАБОТЕ. 
Пояснительная записка содержит:
-титульный лист по образцу, установленному кафедрой,
-задание на специальном бланке,
- содержание,
- введение,
- основная часть,
- заключение,
- список использованной литературы,
- приложения.
Основная часть пояснительной записки содержит:
Описание предметной области (выделение объектов и их связей)
Формализация задачи (математическое описание взаимосвязей между правилами базы знаний (например, с использованием онтологических диаграмм), выбор объектов анализа и математическое описание методов анализа данных)
Описание базы правил и фактов (для экспертной системы)
Описание основных функций (для системы анализа)
Руководство оператора в соответствии с ГОСТ 19.505-79.
Тестирование работы программы.

